
 Sept. 24 2012

Lecture 8: Complex Numbers and Euler's Formula

Reading:

Kreyszig Sections: 13.1, 13.2, , 13.3, 13.4, 13.6

Complex Numbers and Operations in the Complex Plane

Consider, the number zero: it could be *operationally* defined as the number, which when multiplied by any other number always yields itself; and its other properties would follow.

Negative numbers could be defined operationally as something that gives rise to simple patterns. Multiplying by -1 gives rise to the pattern $1, -1, 1, -1, \dots$. In the same vein, a number, i , can be created that doubles the period of the previous example: multiplying by i gives the pattern: $1, i, -1, -i, 1, i, -1, -i, \dots$. Combining the imaginary number, i , with the real numbers, arbitrarily long periods can be defined by multiplication; applications to periodic phenomena is probably where complex numbers have their greatest utility in science and engineering

With $i \equiv \sqrt{-1}$, the complex numbers can be defined as the space of numbers spanned by the vectors:

$$\begin{pmatrix} 1 \\ 0 \end{pmatrix} \text{ and } \begin{pmatrix} 0 \\ i \end{pmatrix} \quad (8-1)$$

so that any complex number can be written as

$$z = x \begin{pmatrix} 1 \\ 0 \end{pmatrix} + y \begin{pmatrix} 0 \\ i \end{pmatrix} \quad (8-2)$$

or just simply as

$$z = x + iy \quad (8-3)$$

where x and y are real numbers. $\text{Re}z \equiv x$ and $\text{Im}z \equiv y$.

Lecture 08 MATHEMATICA® Example 1

Operations on complex numbers

Download [notebooks](#), [pdf\(color\)](#), [pdf\(bw\)](#), or [html](#) from <http://pruffle.mit.edu/3.016-2012>.

Straightforward examples of addition, subtraction, multiplication, and division of complex numbers are demonstrated. An example that demonstrates that MATHEMATICA® doesn't make *a priori* assumptions about whether a symbol is real or complex. An example function that converts a complex number to its polar form is constructed.

<code>imaginary = Sqrt[-1]</code>	1	
<code>(-imaginary)^2</code>	2	1-2: Just like Pi is a <i>mathematical constant</i> , the imaginary number is defined in MATHEMATICA® as something with the properties of <i>i</i>
<small>Complex numbers are composed of a real part + an imaginary part</small>		
<code>z1 = a + i b;</code>	3	3: Here, two numbers that are <i>potentially, but not necessarily</i> complex are defined.
<code>z2 = c + i d;</code>	4	
<code>compadd = z1 + z2;</code>	5	4-5: Addition and multiplication are defined as for any symbol; here the results do not appear to be very interesting <i>because</i> the other symbols could themselves be complex...
<code>compmult = z1 * z2;</code>	6	6: And, Simplify doesn't help much even with assumptions.
<code>Simplify[compmult, a ∈ Reals && b ∈ Reals && c ∈ Reals && d ∈ Reals]</code>	7	7: The real and imaginary parts of a complex entity can be extracted with Re and Im. This demonstrates that MATHEMATICA® hasn't made assumptions about a, b, c, and d.
<small>Mathematica does not assume that symbols are necessarily real...</small>		
<code>Re[compadd]</code>	8	8-12: However, ComplexExpand does make assumptions that symbols are real and, here, demonstrate the rules for addition, multiplication, division, and exponentiation.
<code>Im[compadd]</code>	9	
<small>However, the Mathematica function ComplexExpand does assume that the variables are real...</small>		
<code>ComplexExpand[Re[compadd]]</code>	10	
<code>ComplexExpand[Im[compadd]]</code>	11	
<code>ComplexExpand[Re[z1 / z2]]</code>	12	
<code>ComplexExpand[compmult]</code>	13	13-16: Abs calculates the magnitude (also known as modulus or absolute value) and Arg calculates the argument (or angle) of a complex number. Here, they are used to define a function (<i>Pform</i>) to convert and expression to an equivalent <i>polar form of a complex number</i> .
<code>ComplexExpand[Re[z1^3]]</code>	14	
<code>ComplexExpand[Im[z1^3]]</code>	15	
<small>Function to convert to Polar Form</small>		
<code>Pform[z_] := Abs[z] Exp[i Arg[z]]</code>	16	
<small>Note: the function Arg[z] returns an angle in the range -π to π which measures the inclination of z with respect to the +Re axis in the complex plane.</small>		
<code>Pform[z1]</code>		
<code>Pform[z1 /. {a → 2, b → -π}]</code>		
<code>ComplexExpand[Pform[z1]]</code>		

Complex Plane and Complex Conjugates

Because the complex basis can be written in terms of the vectors in Equation 8-1, it is natural to plot complex numbers in two dimensions—typically these two dimensions are the “complex plane” with $(0, i)$ associated with the y -axis and $(1, 0)$ associated with the x -axis.

The reflection of a complex number across the real axis is a useful operation. The image of a reflection across the real axis has some useful qualities and is given a special name—“the complex conjugate.”




Figure 8-4: Plotting the complex number z in the complex plane: The complex conjugate (\bar{z}) is a *reflection* across the real axis; the minus ($-z$) operation is an *inversion* through the origin; therefore $-(\bar{z}) = (-\bar{z})$ is equivalent to either a reflection across the imaginary axis or an inversion followed by a reflection across the real axis.

The real part of a complex number is the projection of the displacement in the real direction and also the average of the complex number and its conjugate: $\text{Re}z = (z + \bar{z})/2$. The imaginary part is the displacement projected onto the imaginary axis, or the complex average of the complex number and its reflection across the imaginary axis: $\text{Im}z = (z - \bar{z})/(2i)$.

Polar Form of Complex Numbers

There are physical situations in which a transformation from Cartesian (x, y) coordinates to polar (or cylindrical) coordinates (r, θ) simplifies the algebra that is used to describe the physical problem.

An equivalent coordinate transformation for complex numbers, $z = x + iy$, has an analogous simplifying effect for *multiplicative operations* on complex numbers. It has been demonstrated how the

Lecture 08 MATHEMATICA® Example 2

Numerical Properties of Operations on Complex Numbers

Download [notebooks](#), [pdf\(color\)](#), [pdf\(bw\)](#), or [html](http://pruffle.mit.edu/3.016-2012) from <http://pruffle.mit.edu/3.016-2012>.

Several examples demonstrate issues that arise when complex numbers are evaluated numerically.

```

ExactlyOne = Exp[2 π I] 1
NumericallyOne = Exp[N[2 π I]] 2
Chop[NumericallyOne] 3
Round[NumericallyOne] 4
ExactlyI = Exp[π I / 2] 5
NumericallyI = Exp[N[π I / 2]] 6
Round[NumericallyI] 7
Chop[NumericallyI] 8
ExactlyOnePlusI = 9
ComplexExpand[√2 Exp[π I / 4]]
NumericallyOnePlusI = 10
ComplexExpand[√2 Exp[N[π I / 4]]]
Chop[NumericallyOnePlusI] 11
Round[NumericallyOnePlusI] 12
Round[1.5 - 3.5 Sqrt[-1]] 13
Re[NumericallyOnePlusI] 14
Im[NumericallyOnePlusI] 15

```

- 1: The relationship $e^{2\pi i} = 1$ is exact.
- 2: However, $e^{2.0\pi i}$ is *numerically* 1.
- 3: Chop removes small values that are presumed to be the result of numerical imprecision; it operates on complex numbers as well.
- 4: Round is useful for mapping a number to a simpler one in its neighborhood (such as the nearest integer).
- 5–8: Here, the difference between something that is exactly i and is numerically $1.0 \times i$ is demonstrated...
[9–15] And, this is similar demonstration for $1 + i$ using its polar form as a starting point.

Exponentiation and Relations to Trigonometric Functions

Exponentiation of a complex number is defined by:

$$e^z = e^{x+iy} = e^x (\cos y + i \sin y) \quad (8-7)$$

Exponentiation of a purely imaginary number advances the angle by rotation:

$$e^{iy} = \cos y + i \sin y \quad (8-8)$$

combining Eq. 8-8 with Eq. 8-7 gives the particularly useful form:

$$z = x + iy = re^{i\theta} \quad (8-9)$$

and the useful relations (obtained simply by considering the complex plane's geometry)

$$e^{2\pi i} = 1 \quad e^{\pi i} = -1 \quad e^{-\pi i} = -1 \quad e^{\frac{\pi}{2}i} = i \quad e^{-\frac{\pi}{2}i} = -i \quad (8-10)$$

Subtraction of powers in Eq. 8-8 and generalization gives known relations for trigonometric functions:

$$\begin{aligned} \cos z &= \frac{e^{iz} + e^{-iz}}{2} & \sin z &= \frac{e^{iz} - e^{-iz}}{2i} \\ \cosh z &= \frac{e^z + e^{-z}}{2} & \sinh z &= \frac{e^z - e^{-z}}{2} \\ \cos z &= \cosh iz & i \sin z &= \sinh iz \\ \cos iz &= \cosh z & \sin iz &= i \sinh z \end{aligned} \quad (8-11)$$

Complex Numbers in Roots to Polynomial Equations

Complex numbers frequently arise when solving for the roots of a polynomial equation. There are many cases in which a model of system's physical behavior depends on whether the roots of a polynomial are real or imaginary, and if the real part is positive. While evaluating the nature of the roots is straightforward conceptually, this often creates difficulties computationally. Frequently, ordered lists of solutions are maintained and the behavior each solution is followed.

Lecture 08 MATHEMATICA® Example 3

Complex Roots of Polynomial Equations

Download [notebooks](#), [pdf\(color\)](#), [pdf\(bw\)](#), or [html](http://pruffle.mit.edu/3.016-2012) from <http://pruffle.mit.edu/3.016-2012>.

Here we construct an artificial example of a model that depends on a single parameter in a quadratic polynomial and illustrate methods to analyze and visualize its roots. Methods to “peek” at the form of long expressions are also demonstrated.

```

1  sols = Solve[(x^4 - x^3 + x + 1) == 0, x]
2  x /. sols
3  Im[x /. sols]
4  ComplexExpand[Im[x /. sols]]
5  ComplexExpand[Im[x /. sols]] // N
6  ComplexExpand[Re[x /. sols]] // N
   Generalize the above to a family of solutions.
7  bsols = Solve[(x^4 - x^3 + b*x + 1) == 0, x]
8  Dimensions[bsols]
   Short[bsols, 4]
9  SolsbImag = ComplexExpand[Im[x /. bsols]];
10 Dimensions[SolsbImag]
   Short[SolsbImag[[1]]]
11 SolsbReal = ComplexExpand[Re[x /. bsols]];
12 Plot[Evaluate[SolsbImag], {b, -10, 10}]
13 Plot[Evaluate[SolsbImag], {b, -10, 10},
   PlotStyle -> Table[{Hue[1 - a/6]}, {a, 1, 4}]]
14 Plot[Evaluate[SolsbReal], {b, -10, 10},
   PlotStyle -> Table[{Hue[1 - a/6]}, {a, 1, 4}]]
15 Plot[Evaluate[SolsbReal], {b, -10, 10},
   PlotStyle -> Table[{Hue[1 - a/6]},
   Thickness[0.05 - .01*a]}, {a, 1, 4}]]
16 Plot[Evaluate[x /. bsols],
   {b, -10, 10}, PlotStyle -> Thick]

```

- 1–6:** Using a prototype fourth order equation, a list of solutions are obtained; the real and imaginary parts are computed.
- 7:** The above is generalized to a single parameter b in the quartic equation; the conditions that the roots are real will be visualized. `bsols`, the list of solution rule-lists is long and complicated.
- 8:** First, one must consider the structure of `bsols`. `Dimensions` indicates it is a list of four lists, each of length 1. `Dimensions` and `Short` used together, provides a practical method to observe the structure of a complicated expression without filling up the screen display.
- 9–11:** Here, the real and complex parts of *each* of the solutions is obtained with `Re` and `Im` where the parameter b is assumed to be real via the use of `ComplexExpand`. **These may take a long time to evaluate on some computers.**
- 12–13:** Which of the solutions (i.e., 1,2,3, or 4) is identified by a different color (if `Evaluate` is used inside the `Plot` function). In the first case, MATHEMATICA®’s default indexed colors are used, and in the second case they are set explicitly using `Hue` in `PlotStyle`.
- 14:** Similarly, the real parts appear to converge to a single value when the imaginary parts (from above) appear...
- 15:** But, the actual behavior is best illustrated by using `Thickness` to distinguish superimposed values. The behavior of real parts of this solution have what is called a *pitchfork structure*.
- 16:** As of MATHEMATICA® 6, it is not necessary that the plotted function evaluate to a real value at each point. Now, only those points that evaluate to a real number will be graphed.